
1. New parameter for IMS registration functions to distinguish IPCAN and OtherIPCAN
In order to distinguish UTRAN and EUTRA as IPCAN during registration several changes have been done:
a) New Type IMS_IpcanPortType_Type (IMS_PTC_CoordMsg)
	type enumerated IMS_IpcanPortType_Type { IPCAN, OtherIPCAN };


b) IMS_IpcanPortType_Type as additional parameter during IMS registration

	  function f_IMS_REGISTER_InitialRequest(IMS_IpcanPortType_Type p_IpcanType := IPCAN,
                                         template (omit) IMS_DATA_REQ p_IMS_DATA_REQ := omit,

                                         template (omit) charstring p_ExplicitExpiry := omit,

                                         boolean p_SigCompIsActive := false,

                                         boolean p_CondA7 := false,

                                         boolean p_CondA3 := false) runs on IMS_PTC return IMS_DATA_REQ

  {

    var IPCAN_INFO_Type v_IpcanInfo;

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var IMS_RoutingInfo_Type v_RoutingInfo_UL;

    var InternetProtocol_Type v_Protocol;

    var IP_AddrInfo_Type v_UE_Address;

    var IP_AddrInfo_Type v_NW_Address;

    var MessageHeader v_MessageHeader;

    var boolean v_Match := false;

    v_IMS_DATA_REQ := fl_IMS_ReceiveOrGetRegisterRequest(p_IMS_DATA_REQ);

    v_MessageHeader := v_IMS_DATA_REQ.Request.Register.msgHeader;

    v_Protocol := v_IMS_DATA_REQ.RoutingInfo.Protocol;


    select (p_IpcanType) {

      case (IPCAN) {

        v_IpcanInfo := f_IMS_IPCAN_Query(IPCAN);

      }

      case else {

        v_IpcanInfo := f_IMS_IPCAN_Query(OtherIPCAN);

      }

    }
    f_IMS_PTC_SetIpcanInfo(v_IpcanInfo);

    v_Match := match(v_MessageHeader, f_IMS_RegisterRequest_MessageHeaderRX_Initial(v_Protocol));         // @sic R5s130369 sic@

    if (not v_Match) {

      f_IMS_SetVerdictFailOrInconc(__FILE__, __LINE__, "invalid REGISTER Request");

    }

    v_RoutingInfo_UL := v_IMS_DATA_REQ.RoutingInfo;

    v_UE_Address := v_RoutingInfo_UL.UE_Address;

    v_NW_Address := v_RoutingInfo_UL.NW_Address;

    v_Protocol := v_RoutingInfo_UL.Protocol;

    f_IMS_PTC_ImsInfo_Init(v_UE_Address, v_NW_Address, v_IMS_DATA_REQ.Request.Register, p_SigCompIsActive);

    // additional checks

    f_IMS_A_1_1_MessageHeader_CommonChecks(v_MessageHeader, v_Protocol, p_CondA7, v_UE_Address, p_ExplicitExpiry);

    f_IMS_A_1_1_MessageHeader_CheckPublicUserIdentity(v_MessageHeader, p_CondA7, p_CondA3);  // check 'To' and 'From' field (needed for initial registration only); @sic R5s140356: p_CondA3 sic@

    if (p_SigCompIsActive) {

      f_IMS_MessageHeader_CheckSigComp(v_MessageHeader);

    }

    return v_IMS_DATA_REQ;

  }

	  function f_IMS_Register_NonGIBA(IMS_IpcanPortType_Type p_IpcanType := IPCAN,

                                  template (omit) IMS_DATA_REQ p_IMS_DATA_REQ := omit,

                                  charstring p_RegisterExpiration := tsc_IMS_RegisterExpiration,

                                  boolean p_SigCompIsActive := false,

                                  boolean p_IsEmergency := false) runs on IMS_PTC return REGISTER_Request

  { /* Note:

       - In case the the function is called in a default handler the Register message has been already received

       - When used in the IMS CC tests the PTC needs to wait for the reception of the Register message */

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var REGISTER_Request v_RegisterReq;

    v_IMS_DATA_REQ := f_IMS_REGISTER_InitialRequest(p_IpcanType, p_IMS_DATA_REQ, -, p_SigCompIsActive, p_IsEmergency);         // check REGISTER and init GlobalInfo

    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    f_IMS_Register_SecurityInit(v_RegisterReq);

    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo), cs_Response(c_statusLine401, f_IMS_RegisterResponse_401_MessageHeaderTX(v_RegisterReq))));

    v_IMS_DATA_REQ := f_IMS_REGISTER_SubsequentRequest(-, p_IsEmergency);

    v_RegisterReq := v_IMS_DATA_REQ.Request.Register;

    IMS_Server.send(cas_IMS_DATA_RSP(f_IMS_RoutingInfo_ULtoDL(v_IMS_DATA_REQ.RoutingInfo), cs_Response(c_statusLine200, f_IMS_RegisterResponse_200_MessageHeaderTX(v_RegisterReq, p_IsEmergency, p_RegisterExpiration)))); // @sic R5s130561 sic@

    return v_RegisterReq;

  }

	  function f_IMS_Register(IMS_IpcanPortType_Type p_IpcanType := IPCAN,

                          template (omit) IMS_DATA_REQ p_IMS_DATA_REQ := omit,

                          boolean p_SigCompIsActive := false,

                          charstring p_RegisterExpiration := tsc_IMS_RegisterExpiration) runs on IMS_PTC return REGISTER_Request

  { /* @sic R5s130756 additional changes: p_IsGIBA removed sic@ */

    /* @sic R5-151769 additional parameter p_RegisterExpiration sic@ */

    var REGISTER_Request v_RegisterRequest;

    if (f_IMS_PTC_SecurityScheme_IsGiba()) {

      if (p_SigCompIsActive) {

        FatalError(__FILE__, __LINE__, "sigcomp is not supported for GIBA");

      }

      v_RegisterRequest := f_IMS_Register_GIBA(p_IMS_DATA_REQ, p_RegisterExpiration);

    } else {

      v_RegisterRequest := f_IMS_Register_NonGIBA(p_IpcanType, p_IMS_DATA_REQ, p_RegisterExpiration, p_SigCompIsActive);

    }

    return v_RegisterRequest;

  }

	  function f_IMS_Registration(IMS_IpcanPortType_Type p_IpcanType := IPCAN,

                              template (omit) IMS_DATA_REQ p_IMS_DATA_REQ := omit,

                              boolean p_SigCompIsActive := false,

                              charstring p_RegisterExpiration := tsc_IMS_RegisterExpiration) runs on IMS_PTC

  {

    var REGISTER_Request v_RegisterRequest;

    var boolean v_WaitForPublish := true;   /* @sic R5-155916: handling PUBLISH during registration sic@ */

    v_RegisterRequest := f_IMS_Register(p_IpcanType, p_IMS_DATA_REQ, p_SigCompIsActive, p_RegisterExpiration);

    f_IMS_Register_SubscribeNotify(v_RegisterRequest, v_WaitForPublish);

  }


c) Further changes: 

Default value ('-') used for other calls of f_IMS_REGISTER_InitialRequest, f_IMS_Register_NonGIBA, f_IMS_Register and f_IMS_Registration

2. Change of IMS_DefaultHandler
a) Trigger from OtherIPCAN to allow IMS registration on other IPCAN (UTRAN)

b) Use of IMS_IpcanPortType_Type to distinguish IPCAN and OtherIPCAN

	  function f_IMS_DefaultHandler() runs on IMS_PTC

  {

    var boolean v_IsRegistered := false;

    var IMS_DATA_REQ v_IMS_DATA_REQ; 
    var IMS_IpcanPortType_Type v_IpcanType := IPCAN;
    f_IMS_PTC_Init(-, defaultHandling, IMS_Security);

    while (true) {

      alt { 
        [] OtherIPCAN.receive(cmr_IMS_IPCAN_Trigger) {

          v_IpcanType := OtherIPCAN;

        }
        [] IPCAN.receive(cmr_IPCAN_IMS_Reset) {

          v_IsRegistered := false; 
          v_IpcanType := IPCAN;
        }

        [] IMS_Server.receive(car_IMS_Register_Request(?)) -> value v_IMS_DATA_REQ {      // receive any REGISTER

          if (not v_IsRegistered) {

            f_IMS_DefaultHandler_InitialRegistration(v_IpcanType, v_IMS_DATA_REQ);  // @sic R5s140788 change 2 sic@

            v_IsRegistered := true;

          } else {

            f_IMS_PTC_ResetRanType();                 // @sic R5s150057 change 3: reset RanType to allow de-registrtion from any RAT sic@

            f_IMS_PTC_Deregistration(v_IMS_DATA_REQ); // @sic R5-132037 sic@

            /* @sic R5s140219 - MCC160 implementation: closing TCP connections is handled in f_IMS_PTC_Deregistration sic@ */

            v_IsRegistered := false;

          }

        }

      }

      if (not v_IsRegistered) {  /* the UE is not registered anymore => reset IMS

                                    @sic R5s150231 change 2 sic@ */

        f_IMS_GlobalRelease();   /* reset IMS_PTC and IP_PTC's registration info in case of de-registration or the UE being switched off (-> cmr_IPCAN_IMS_Reset) */

      }

    }

  }

	  function f_IMS_DefaultHandler_InitialRegistration(IMS_IpcanPortType_Type p_IpcanType := IPCAN,

                                                    template (omit) IMS_DATA_REQ p_IMS_DATA_REQ := omit) runs on IMS_PTC

  {
    fl_IMS_DefaultHandler_SendCoOrdMsg(p_IpcanType);
// send trigger to EUTRA PTC to indicate start of IMS registration @sic R5-123731 sic@

    f_IMS_Registration(p_IpcanType, p_IMS_DATA_REQ);

    f_IMS_TCP_CloseHandling(justWait);   /* @sic R5s140219 - MCC160 implementation sic@ */
    fl_IMS_DefaultHandler_SendCoOrdMsg(p_IpcanType);
// send trigger to EUTRA PTC to indicate end of IMS registration @sic R5-123731 sic@

  }

	  function fl_IMS_DefaultHandler_SendCoOrdMsg(IMS_IpcanPortType_Type p_IpcanType := IPCAN) runs on IMS_PTC

  {

    select (p_IpcanType) {

      case (IPCAN) {

        f_IMS_IPCAN_SendCoOrdMsg(IPCAN);

      }

      case else {

        f_IMS_IPCAN_SendCoOrdMsg(OtherIPCAN);

      }

    }

  }


3. Enhancement of f_UTRAN_DelayForUserPlaneSignalling
a) New Type DelayForUserPlaneSignalling_Type (IMS_PTC_CoordMsg) replacing EUTRA_DelayForUserPlaneSignalling_Type and UTRAN_DelayForUserPlaneSignalling_Type)

	  type enumerated DelayForUserPlaneSignalling_Type { noDelay, waitForIMS, dontWaitForIMS };


Affects

· f_EUTRA_IdleUpdated_Step16_17
· f_EUTRA_IdleUpdated_ConnectToAdditionalPDN
· f_EUTRA_DelayForUserPlaneSignalling
· f_EUTRA_NAS_AttachComplete
· f_UTRAN_DelayForUserPlaneSignalling
· f_UTRAN_UE_64kPS_Rb_Est

· f_UTRAN_UE_FACH_PS_Rb_Est
b) New common function f_IPCAN_DelayForUserPlaneSignalling (IMS_PTC_CoordMsg) to be used by f_EUTRA_DelayForUserPlaneSignalling and f_UTRAN_DelayForUserPlaneSignalling
	  function f_IPCAN_DelayForUserPlaneSignalling(IMS_IPCAN_CO_ORD_PORT p_Port,

                                               DelayForUserPlaneSignalling_Type p_DelayForUserPlaneSignalling)

  {

    var boolean v_WaitForFirstTrigger := true;

    timer t_WaitForUPlaneSignalling;

    select (p_DelayForUserPlaneSignalling) {

      case (noDelay) {

        /* do nothing */

      }

      case (dontWaitForIMS) {

        t_WaitForUPlaneSignalling.start(1.2);

        alt {

          [] t_WaitForUPlaneSignalling.timeout {

            // just continue

          }

          [] p_Port.receive(cmr_IMS_IPCAN_Trigger) {

            f_SetVerdictInconc(__FILE__, __LINE__, "IMS Registration even though loopback mode is activated");  /* @sic R5-153971 sic@ */

          }

        }

      }

      case (waitForIMS) {

        t_WaitForUPlaneSignalling.start(10.0);

        alt {

          [] t_WaitForUPlaneSignalling.timeout {

            // just continue

          }

          [] p_Port.receive(cmr_IMS_IPCAN_Trigger) {

            if (v_WaitForFirstTrigger) {                // @sic R5s120727 change 1 sic@

              t_WaitForUPlaneSignalling.stop;

              v_WaitForFirstTrigger := false;

              repeat;      // receive second trigger sent by IMS after registration has finished

            }

          }

        }

      }

    }

  }

	  function f_EUTRA_DelayForUserPlaneSignalling(DelayForUserPlaneSignalling_Type p_DelayForUserPlaneSignalling:= waitForIMS) runs on EUTRA_PTC

  {


    var boolean v_LoopbackABIsActivated;
 
    var DelayForUserPlaneSignalling_Type v_DelayForUserPlaneSignalling := p_DelayForUserPlaneSignalling;

    if (p_DelayForUserPlaneSignalling == waitForIMS) {

      v_LoopbackABIsActivated := f_EUTRA_LoopbackABIsActivated_Get();  /* @sic R5-153971 sic@ */

      if (not pc_IMS or v_LoopbackABIsActivated) {

        v_DelayForUserPlaneSignalling := dontWaitForIMS;

      }

    }
    f_IPCAN_DelayForUserPlaneSignalling(IMS[tsc_Index_PDN1], p_DelayForUserPlaneSignalling);



























  }

	  function f_UTRAN_DelayForUserPlaneSignalling(DelayForUserPlaneSignalling_Type p_WaitDataSignalling := noDelay)

  { 
    f_IPCAN_DelayForUserPlaneSignalling(IMS[tsc_Index_PDN1], p_WaitDataSignalling);  



  }


4. Handling of IPCAN_QUERY in UTRAN

IPCAN_QUERY needs to be handled at the UTRAN_PTC. But in contrast to EUTRA this handling cannot be added to the standard default behaviour as this is common for LTE and UTRAN test suites (including SSNITZ), where there are no coordination ports to IMS.
( Handling of IPCAN_QUERY in UTRAN shall be done by a local default behaviour which is activated just for the registration (see below)
	  altstep a_UTRAN_IMS_QueryHandler(IMS_IPCAN_CO_ORD_PORT p_ImsCoorPort) runs on UTRAN_PTC

  {

    [] p_ImsCoorPort.receive(cmr_IMS_IPCAN_Command(IPCAN_QUERY))

      {

        p_ImsCoorPort.send(cms_IPCAN_IMS_QueryResponse(f_UTRAN_IMS_IpcanInfo()));
        repeat;
      }

  }


Note: f_UTRAN_IMS_IpcanInfo and related functions are moved from IPCAN_UTRAN to UTRAN_IRAT_CommonProcedures.

5. Changes for UTRAN PS radio bearer establishment functions

a) Trigger to be sent to IMS_DefaultHandler to use OtherIPCAN (UTRAN) during IMS registration

b) Default handler needs to be activated to deal with IPCAN_QUERY during IMS registration

c) New function for common handling of 64k RB and FACH RB
	  function f_UTRAN_UE_PsRB_Est_InitialRegistration(UTRAN_CellId_Type p_CellId,

                                                   RB_ConfigType p_RB_Configuration,

                                                   DelayForUserPlaneSignalling_Type p_DelayForUserPlaneSignalling) runs on UTRAN_IRAT_PTC

  {

    var DelayForUserPlaneSignalling_Type v_DelayForUserPlaneSignalling := p_DelayForUserPlaneSignalling;

    var default v_DefaultVar := null;

    if (p_DelayForUserPlaneSignalling == waitForIMS) {

      if (pc_IMS) {

        v_DefaultVar := activate(a_UTRAN_IMS_QueryHandler(IMS[tsc_Index_PDN1]));

        f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_PDN1]);     // send trigger to inform IMS_DefaultHandler that 'OtherIPCAN' shall be used as coordination port during IMS registration 

      } else {

        v_DelayForUserPlaneSignalling := dontWaitForIMS;

      }

    }

    f_UTRAN_UE_PsRB_Est(p_CellId, p_RB_Configuration);

    // Optional 1.2s delay acc. to 34.108 cl. 7.2.6 to allow user-plane signalling

    f_UTRAN_DelayForUserPlaneSignalling(v_DelayForUserPlaneSignalling); // @sic R5s130384 sic@

    deactivate(v_DefaultVar);

  }

	  function f_UTRAN_UE_64kPS_Rb_Est(UTRAN_CellId_Type p_CellId,

                                   DelayForUserPlaneSignalling_Type p_WaitDataSignalling := noDelay) runs on UTRAN_IRAT_PTC

  { 
    f_UTRAN_UE_PsRB_Est_InitialRegistration(p_CellId, cell_DCH_64kPS_RAB_SRB, p_WaitDataSignalling);



  }

	  function f_UTRAN_UE_FACH_PS_Rb_Est(UTRAN_CellId_Type p_CellId,

                                     DelayForUserPlaneSignalling_Type p_WaitDataSignalling := noDelay) runs on UTRAN_IRAT_PTC

  { 
    f_UTRAN_UE_PsRB_Est_InitialRegistration(p_CellId, cell_FACH_PS, p_WaitDataSignalling);



  }


6. Affected test case functions: waitForIP replaced by waitForIMS
f_UTRAN_UE_FACH_PS_Rb_Est(utran_Cell5, waitForIP) changed to f_UTRAN_UE_FACH_PS_Rb_Est(utran_Cell5, waitForIMS) in

· f_TC_6_2_4_X_UTRAN
· f_TC_6_2_4_2_UTRAN
· f_TC_6_2_4_5_UTRAN
· f_TC_6_2_4_7_UTRAN
· f_TC_6_2_3_4_UTRAN
· f_TC_6_2_3_4a_UTRAN
· f_TC_9_2_1_2_1b_UTRAN

· f_TC_9_2_1_2_1c_UTRAN
